Formularinhalte dynamisch ändern Teil 1
Versionen: Access 2003, 2002/XP, 2000 und 97

Die Datenbasis von Formularen legen Sie normalerweise über Abfragen fest. Was aber, wenn Sie zum Beispiel nur einmal schnell alle Kunden aus einem bestimmten Ort oder alle Artikel einer Artikelgruppe anzeigen und durchgehen möchten? Natürlich können Sie dazu verschiedene Abfragen anlegen und diese mit unterschiedlichen Formularen öffnen. Access bietet aber weitaus bequemere Möglichkeiten. In diesem Tipp der Woche zeigen wir Ihnen, welche das sind und wie Sie sie effektiv einsetzen.

Ausgangspunkt für die Veränderung von Formularinhalten ist die Eigenschaft "RecordSource" des Formulars. Über sie legen Sie fest, welche Datensätze das Formular überhaupt anzeigt. Der Eigenschaft ist in der Regel eine Tabelle, eine Abfrage oder eine Selektion per SQL wie beispielsweise "select * from Kunden" zugewiesen. Um die Datenbasis, also den Inhalt von Formularen, zu ändern, können Sie also einmal die Eigenschaft "RecordSource" ändern.

Zum Beispiel: Im Formular für Ihre Kundenadressen möchten Sie eine Schaltfläche "Kunden im gleichen Ort" implementieren. Nach einem Klick auf die Schaltfläche soll das Formular alle Kunden anzeigen, die im gleichen Ort wie der aktuell angezeigte Kunde wohnen. Eine solche Funktion ist beispielsweise für Außendienstmitarbeiter recht sinnvoll, die so etwas einfacher ihre Besuche planen können. Anhand des Formulars "Kunden" der Beispieldatenbank "Nordwind" soll der Einbau einer solchen Funktion einmal demonstriert werden. Legen Sie dort zunächst eine Schaltfläche "btnGleicherInhalt" oder ähnlich an, für deren Eigenschaft "Beschriftung" Sie "Kunden im gleichen Ort" eingeben. In die Ereignis-Prozedur "Beim Klicken" der Schaltfläche kopieren Sie folgende Anweisungen:

Private Sub btnGleicherInhalt_Click()
  Dim strSQL As String

  If Me.NewRecord Then
    Beep
    Exit Sub
  End If

  strSQL = "SELECT * FROM Kunden " & _
           "WHERE [Ort] = '" & Me.Ort & "'"

  Me.RecordSource = strSQL

End Sub

Hier prüfen wir zunächst über die Eigenschaft "NewRecord" , ob gerade ein neuer Datensatz erfasst wird. Wenn ja, gibt es einen kleinen akustischen Hinweis und die Routine wird verlassen. Andernfalls setzen wir eine neue SQL-Abfrage zusammen, die alle Datensätze selektiert, bei denen das Feld "Ort" den gleichen Inhalt hat, wie beim aktuell angezeigten Kunden. Die Zuweisung "Me.RecordSource = strSQL" sorgt abschließend dafür, dass die Datenbasis des Formulars geändert und die Anzeige neu aufgebaut wird. Den Erfolg erkennen Sie an der Änderung des Status in den Navigationselementen - hier wird nun zum Beispiel "3 von 5" angezeigt. Es hat also sozusagen eine Filterung der Datensätze stattgefunden

Um wieder alle Datensätze im Formular verfügbar zu haben, muss eine weitere Schaltfläche vorhanden sein, die die Datenbasis entsprechend zurücksetzt. Ergänzen Sie das Formular um eine kleine Schaltfläche "btnAll" mit dem über die Eigenschaft "Bild" eingestellten "Filter aufheben"-Symbol direkt vor der Schaltfläche Kunden im gleichen Ort.

In die Ereignis-Prozedur "Beim Klicken" kopieren Sie folgende Anweisung:

Private Sub btnAll_Click()

  Me.RecordSource = "SELECT * FROM Kunden"

End Sub

Durch diese Zuweisung werden alle Datensätze der Tabelle "Kunden" selektiert und als Datenbasis zugeordnet. Anschließend kann also wieder durch alle Datensätze geblättert werden.

Für die Einbindung einer solchen Lösung in eigene Projekte hier noch einmal das Vorgehen Schritt für Schritt:

1. Öffnen Sie das betreffende Formular im Entwurfsmodus.

2. Legen Sie zwei Schaltflächen "btnAll" oder ähnlich mit dem Symbol "Filter aufheben" und "btnGleicherInhalt" oder ähnlich mit entsprechender Beschriftung an. Möchten Sie beispielsweise alle Artikel einer Artikelgruppe anzeigen lassen, wäre die Beschriftung "Artikel dieser Gruppe" oder ähnlich.

3. Stellen Sie jeweils die Eigenschaft "Beim Klicken" auf den Eintrag "Ereignisprozedur" und klicken Sie auf die Schaltfläche mit den drei Punkten dahinter, um den VBA-Editor zu öffnen.

4. Geben Sie die oben gezeigten VBA-Routinen für die jeweilige Schaltfläche ein. Statt des Feldnamens "Ort" verwenden Sie das Feld, das für die Selektion als Kriterium herangezogen werden soll, also zum Beispiel "ArtikelGruppe". Denken Sie auch daran, die Tabelle entsprechend anzugeben, also beispielsweise "Artikel". Das könnte dann zum Beispiel so aussehen:

Private Sub btnGleicheGruppe_Click()
  Dim strSQL As String

  If Me.NewRecord Then
    Beep
    Exit Sub
  End If

  strSQL = "SELECT * FROM Artikel " & _
           "WHERE [ArtikelGruppe] = '" _
           & Me.ArtikelGruppe & "'"

  Me.RecordSource = strSQL

End Sub

Private Sub btnAll_Click()

  Me.RecordSource = "SELECT * FROM Artikel"

End Sub

Im zweiten Teil dieses Tipps der Woche in der nächsten Ausgabe von Access Weekly zeigen wir eine weitere Möglichkeit, den Inhalt von Formularen dynamisch zu steuern.

Die im ersten Teil dieses Tipps der Woche beschriebene Lösung ist recht praktikabel, solange "einfache" Formulare zum Einsatz kommen. Häufig liegen Formularen aber komplexe Abfragen zugrunde, die die anzuzeigenden Datensätze aus mehreren Tabellen "zusammensuchen" und dabei auch nur bestimmte Felder selektieren. Hier die passenden SQL-Abfragen zusammenzusetzen, ist mühsam und fehlerträchtig. Zum Glück stellt Access weitere Mechanismen bereit, die das Filtern von Datensätzen erheblich vereinfachen. Basis sind die Eigenschaften "Filter" und "FilterOn".

Die Eigenschaft "Filter" legt zunächst fest, nach welchem Kriterium die Datensätze selektiert werden sollen. Hierbei handelt es sich um eine SQL-Abfrage, die allerdings lediglich aus dem "Where"-Teil besteht:

[Land]= '" & Me.Land & "'"

Damit legen Sie aber nur das Kriterium fest. Um die Filterung zu aktivieren, muss noch folgende Anweisung gegeben werden:

FilterOn = True

Analog dazu schaltet

FilterOn = False

die Filterung wieder aus. Dieser Mechanismus hat den Vorteil, dass die zugrunde liegende Abfrage durchgehend verwendet wird, man sich also nicht um die Selektion einzelner Felder aus einer oder mehreren Tabellen kümmern muss.

Das wird am einfachsten wieder an einem praktischen Beispiel deutlich: Im Formular "Kunden" der Beispieldatenbank "Nordwind" soll eine Schaltfläche Kunden im gleichen Land für die Filterung über die Eigenschaft "Filter" sowie eine zusätzliche Schaltfläche zur Aufhebung des Filters über "FilterOn" eingebaut werden.

Schauen wir zunächst, wie die Filterung über eine Schaltfläche gesetzt wird:

Private Sub btnGleicherInhalt_Click()
    Dim strFilter As String

    If Me.NewRecord Then
        Beep
        Exit Sub
    End If

    strFilter = "[Land] = '" & Me.Land & "'"
    Me.Filter = strFilter
    Me.FilterOn = True

End Sub

Zuerst prüfen wir wieder, ob gerade ein neuer Datensatz erfasst wird. Wenn ja: Akustischer Hinweis, Ende der Aktion. Andernfalls setzen wir in der Variablen "strFilter" die Where-Klausel für die Selektion zusammen, weisen sie der Eigenschaft "Filter" zu und aktivieren die Filterung über "FilterOn = True". Um die Filterung wieder aufzuheben ist auch hier nur eine einzige Anweisung notwendig:

Private Sub btnFilterOff_Click()

    Me.FilterOn = False

End Sub

Um eine solche Lösung in eigene Projekte einzubinden, gehen Sie wie folgt vor:

1. Öffnen Sie das betreffende Formular im Entwurfsmodus.

2. Legen Sie zwei Schaltflächen "btnFilterOff" oder ähnlich mit dem Symbol "Filter aufheben" und "btnGleicherInhalt" oder ähnlich mit entsprechender Beschriftung an. Möchten Sie beispielsweise alle Artikel einer Artikelgruppe anzeigen lassen, wäre die Beschriftung Artikel dieser Warengruppe oder ähnlich.

3. Stellen Sie jeweils die Eigenschaft "Beim Klicken" auf den Eintrag "Ereignisprozedur" und klicken Sie auf die Schaltfläche mit den drei Punkten, um den VBA-Editor zu öffnen.

4. Geben Sie die oben gezeigten VBA-Routinen für die jeweilige Schaltfläche ein. Statt des Feldnamens "Land" verwenden Sie das Feld, das für die Selektion als Kriterium herangezogen werden soll, also zum Beispiel "ArtikelGruppe". Denken Sie auch daran, die Tabelle entsprechend anzugeben, also zum Beispiel "Artikel":

Private Sub btnGleicheGruppe_Click()
    Dim strFilter As String

    If Me.NewRecord Then
        Beep
        Exit Sub
    End If

    strFilter = "[ArtikelGruppe] = '" & _
                Me.ArtikelGruppe & "'"
    Me.Filter = strFilter
    Me.FilterOn = True

End Sub

